

Algorithm 435

Modified Incomplete Gamma Function [S 14]

Wayne Fullerton [Recd. 30 Dec. 1970 and 12 April 1971]

Department of Astronomy, University of Michigan, Ann Arbor, MI 48104

Key Words and Phrases: modified incomplete Gamma function, incomplete Gamma function, chi-square distribution function, Poisson distribution function

CR Categories: 5.13

Language: Fortran

Description

The incomplete Gamma function is defined by

$$\gamma(a, x) = \int_0^x y^{a-1} \cdot e^{-y} dy, \quad x \geq 0. \quad (1)$$

If x is allowed to assume negative values and if the absolute value of y is substituted for y in the term y^{a-1} , then a modified incomplete Gamma function may be defined by

$$\gamma'(a, x) = \int_0^x |y|^{a-1} \cdot e^{-y} dy, \quad -\infty < x \leq \infty. \quad (2)$$

Note that if x is less than zero, the above is equivalent to

$$\gamma'(a, x) = - \int_0^{|x|} y^{a-1} \cdot e^{+y} dy, \quad x \leq 0. \quad (3)$$

The function subprogram *GAMINC* given below computes the more general function

$$\begin{aligned} \text{GAMINC}(a, x_1, x_2) &\cong e^{x_1} \int_{x_1}^{x_2} |y|^{a-1} \cdot e^{-y} dy \\ &= e^{x_1} [\gamma'(a, x_2) - \gamma'(a, x_1)]. \end{aligned} \quad (4)$$

For x_1 equal to zero, *GAMINC* is just a modified incomplete Gamma function. And if x_2 is also greater than or equal to zero, then *GAMINC* is simply an incomplete Gamma function.

The need for the function *GAMINC* arises in the calculation of

$$I \equiv \int_{z_1}^{z_2} e^{a+bz} \exp \left\{ - \int_0^z e^{a+\beta z'} \frac{dz'}{-\sin(\vartheta)} \right\} \frac{dz}{-\sin(\vartheta)}, \quad (5)$$

where ϑ is an angle between $-\pi$ and $+\pi$ not equal to zero. The two constants b and β are of the same sign. The integral in the exponent

The facilities of the University of Michigan Computing Center were used for this work and the research was supported by the National Science Foundation through a Traineeship granted to the author. Author's present address: Los Alamos Scientific Laboratory, Los Alamos, NM 87544.

can be done explicitly to yield

$$I = \frac{e^{a-\tau_0+x_1}}{-\sin(\vartheta)} \int_{z_1}^{z_2} e^{a+bz} \exp \left\{ \frac{-e^{a+\beta z}}{-\sin(\vartheta)} \right\} dz, \quad (6)$$

where

$$X_i = \frac{e^{a+bz_i}}{-b \sin(\vartheta)}$$

and

$$\tau_0 = \int_0^{z_1} e^{a+\beta z'} \frac{dz'}{-\sin(\vartheta)}$$

A change of variables finally reduces the above integral to

$$I = e^{a-\tau_0} |b \sin(\vartheta)|^{\beta/b-1} \cdot e^{-a\beta/b} \left[e^{x_1} \int_{x_1}^{x_2} |y|^{\beta/b-1} \cdot e^{-y} dy \right] \quad (7)$$

The quantity in brackets is *GAMINC*($\beta/b, X_1, X_2$).

The approximations of $\gamma'(a, x)$ used in *GAMINC* are valid only for $1. \lesssim a \lesssim 2$. (See Table I.) The user may compute *GAMINC* for other values of a with the aid of the recurrence relation (m is a positive integer such that $1. \lesssim a \lesssim 2$).

$$\begin{aligned} \text{GAMINC}(m+a, x_1, x_2) \\ = (m+a-1) \text{GAMINC}(m+a-1, x_1, x_2) + [|x_1|^{m+a-1} \\ - |x_2|^{m+a-1} e^{x_1-x_2}] \end{aligned} \quad (8)$$

In general for $x_1 \geq 0$ and $x_2 \geq 0$,

$$\begin{aligned} \text{GAMINC}(m+a, x_1, x_2) \\ = (m+a-1) \cdot (m+a-2) \cdots (a) \cdot \text{GAMINC}(a, x_1, x_2) \\ + |x_1|^a [|x_1|^{m-1} + \sum_{i=1}^{m-1} (m+a-1) \\ \cdots (m+a-i) |x_1|^{m-1-i}] \\ - |x_2|^a [|x_2|^{m-1} + \sum_{i=1}^{m-1} (m+a-1) \\ \cdots (m+a-i) |x_2|^{m-1-i}] e^{x_1-x_2}. \end{aligned} \quad (9)$$

The recurrence relation should be applied in the other direction if $m+a$ is less than 1.

For large values of a ($a \gtrsim 15$) in the incomplete Gamma function, the user is referred to the algorithm by Takenaga [5].

In all cases we use approximations which are functions of both a and x , so that it is not necessary to compute and store an economized polynomial for each value of a . The overhead in execution time for doing this is not significant since many-term expressions would result anyway. Also exponentiation and real numbers raised to a real power require 30 percent of the total computing time. Multiplying $\gamma'(a, x_2) - \gamma'(a, x_1)$ by e^{x_1} saves two exponentiations and greatly extends the range over which the difference can be represented without over- or underflows occurring. Four separate approximations are used to compute $\gamma'(a, x)$.

Region 1. For $x \geq 5.0$, the complementary incomplete Gamma function is computed by using a continued fraction approximation [1]

$$\Gamma(a) - \gamma'(a, x) = \frac{e^{-x} x^a}{x + T_1}, \quad (10)$$

where

$$T_i = \frac{i+a}{1+i/(x+T_{i+1})},$$

and where $\Gamma(a)$ is the complete Gamma function of a . Only terms through T_3 are used explicitly. T_4 is taken into account in an approximate way by setting $T_4 = 1.7$, which is its approximate value when $x \sim 5.0$. If both argument values are greater than 5.0, then significance is maintained by subtracting the complementary functions, not the functions themselves.

Region 2A. For $-12. < x < -1.$ and $1. < x < 5.,$ the continued fraction approximation given by Luke [3] is valid. We rewrite the approximation in the form

$$\gamma'(a, x) = \frac{x \cdot |x|^{a-1} \cdot e^{-x}}{a \cdot T_1}, \quad (11)$$

where

$$T_n = 1. - \frac{(a + n - 1) \cdot x}{(a + 2n - 2) \cdot [a + 2n - 1 + (n \cdot x) / ((a + 2n) \cdot T_{n+1})]}.$$

Only terms through T_7 are used explicitly, and T_8 is computed by using the approximate expression

$$T_8 \cong 1.00150 - 8.95 \cdot 10^{-8} \cdot a + x \cdot (-0.0337062 + .0004182 \cdot a + x \cdot (.000999294 - .000104103 \cdot a)). \quad (12)$$

On a computer with 32 bit words, eq. (11) must be evaluated in double precision in order to maintain approximately six significant figures of accuracy. On an IBM 360 double precision evaluation can be forced by including more than seven digits in a constant as is done in eq. (12). Of course, double precision evaluation is unnecessary if there are somewhat more than 32 bits per word. Because the calculation of the approximation of $\gamma'(a, x)$ is a relatively time consuming operation, a separate approximation is used when $|x| \leq 1.$

Region 2B. For $-1.0 \leq x \leq 1.0,$ a change of variables is made so that

$$\gamma'(a, x) = |x|^{a-1} \cdot e^{-x} \int_0^x \left(\frac{y}{x}\right)^{a-1} \cdot e^{-y+x} dy, \quad (13)$$

or

$$\gamma'(a, x) = x \cdot |x|^{a-1} \cdot e^{-x} \int_0^1 (1-p)^{a-1} \cdot e^{xp} dp. \quad (14)$$

Because $-1.0 \leq xp \leq 1.0,$ e^{xp} may be adequately approximated with a polynomial. A Chebyshev approximation of nine terms yields a maximum absolute error less than $10^{-7},$ which is adequate to insure that the maximum relative error of the integral ordinarily be much less than about $10^{-6}.$ Since the relative error in the single precision evaluation of $|x|^{a-1} \cdot e^{-x}$ is usually $\sim 1 \cdot 10^{-6}$ for a machine with a 32 bit word length, the above error bound seems entirely reasonable. Write

$$e^Z \cong \sum_{i=0}^M b_i Z^i, \quad -1.0 \leq Z \leq 1.0. \quad (15)$$

Then

$$\gamma'(a, x) \cong x \cdot |x|^{a-1} \cdot e^{-x} \sum_{i=0}^M \frac{i! b_i x^i}{(i+a)(i+a-1) \cdots (a)}. \quad (16)$$

Finally we may define $b_i' = b_i \cdot i!,$ and write

$$\gamma'(a, x) \cong x \cdot |x|^{a-1} \cdot e^{-x} \sum_{i=0}^M \frac{b_i' x^i}{(i+a) \cdot (i+a-1) \cdots (a)}. \quad (17)$$

Note that if the series was not economized, all the b_i' would be unity. But because a finite Chebyshev economized series is employed, the b_i' are only approximately unity.

Of course, it would be possible to extend the Chebyshev approximation to include the entire range $-12. < x < 5.0;$ however the many-term result would have to be evaluated in double precision in order to insure a relative error $< 10^{-6}.$ It would also be possible to decrease the range of validity of the ascending continued fraction approximation; however the other approximations would then have to be more complicated and would require an accordingly longer time to evaluate. Such a change was judged inadvisable since the function is used predominantly with arguments whose absolute

Table I. Relative Errors of $GAMINC(A, 0., X)$ in Units of the Sixth Decimal Place

X	A								
	0.5	0.8	1.1	1.4	1.7	2.0	2.3	2.9	3.5
-14.	16.57	3.37	0.27	0.79	1.11	0.81	1.87	1.35	1.03
-12.	1.03	1.61	0.42	0.47	1.23	0.95	2.70	1.69	2.70
-8.	0.09	1.04	0.34	0.69	1.69	1.07	2.01	2.11	2.24
-4.	0.38	0.82	0.15	0.79	0.67	0.19	0.51	1.59	0.75
-2.	1.22	0.78	0.50	0.11	0.07	0.01	0.09	0.11	0.51
-0.5	0.65	0.43	0.21	0.32	0.23	0.15	0.18	0.65	0.78
+0.5	1.03	0.42	0.66	1.19	1.25	0.77	0.29	0.06	0.35
2.	0.53	1.57	0.77	0.26	0.04	0.22	0.05	0.23	0.40
6.	0.44	0.38	0.06	0.01	0.06	0.06	1.21	0.36	2.21
10.	0.63	0.73	0.03	0.02	0.08	0.03	0.60	0.04	0.26

Table II. Execution Times of $GAMINC(A, X_1, X_2)$ in Milli-seconds

	$X_2 \leq -12.$	$-12. < X_2 < 5$		$X_2 \geq 5.$
		$ X_2 \leq 1.$	$ X_2 > 1.$	
$X_1 \leq -12.$	1.1	1.4	2.0	0.6*
$-12. < X_1 < 5. \begin{cases} X_1 \leq 1. \\ X_1 > 1. \end{cases}$	1.4	1.3	2.0	1.4
		2.0	2.4	2.0
$X_1 \geq 5.$	0.6*	1.4	2.0	1.1
$X_1 = 0.$	0.8	0.9	1.4	0.8

* Only the modified incomplete Gamma function for $X = X_1$ was calculated, because $|X_2 - X_1|$ was greater than *EXPLIM*.

values are large. Also, the present choice of ranges and approximations provides for the accurate representation of $\gamma'(a, x)$ further beyond $a = 2.$ than would many other choices.

Region 3. For $x \leq -12.,$ the asymptotic expansion

$$\gamma'(a, x) \approx \Gamma(a) - |x|^{a-1} \cdot e^{-x} \left[1 + \frac{a-1}{x} + \frac{(a-1) \cdot (a-2)}{x^2} + \cdots \right] \quad (18)$$

is used. Shank's e_1 process [4] is applied once to the six-term series in order to accelerate convergence.

The function subprogram is invoked by a reference of the form $GAMINC(A, X_1, X_2, GAM),$

where GAM is the user-supplied value of the complete Gamma function of $A.$ $\Gamma(a)$ is now commonly a part of the standard Fortran library of functions. If it is not, one of the several algorithms described in this department may be used, or *GAMMA* given in IBM's Scientific Subroutine Package (cf. Hastings [2]) may be used.

Table I presents the absolute value of the relative errors (multiplied by 10^6) of $\gamma'(a, x)$ for selected values of a and $x.$ Because $|x|^{a-1} \cdot e^{-x}$ was not calculated in double precision, these errors are the total errors and not the errors in the approximations. The "exact" values were found by directly summing the series

$$\gamma'(a, x) \cong |x|^a \sum_{i=0}^N \frac{(-x)^i}{(a+i)!}$$

in double precision. N was chosen so that the contribution of the N th term was less than $2 \cdot 10^{-9}$ times the sum of the previous N terms. Single precision approximations were used to represent a and x in order to insure that the series and the subprogram gave $\gamma'(a, x)$ for the same parameter values. The subroutine has been used extensively to compute a three-fold integral which includes numerous cases of eq. (5) as a part of the integral. Independent numerical integration results are in agreement with subroutine results to within three significant figures—the accuracy of the numerical integration. Table II gives the average execution times in milliseconds of the subroutine for various argument combinations. The times are for an IBM 360/67, which, for comparison, exponentiates in approximately 0.1 milliseconds.

Acknowledgments. It is a pleasure to thank Dr. Carl deBoor for commenting on a draft of this paper.

References

1. Abramowitz, M., and Stegun, I.A. *Handbook of Mathematical Functions*. National Bureau of Standards, U.S. Gov. Print. Off., Washington, D.C., 1967, p. 263.
2. Hastings, C. *Approximations for Digital Computers*. Princeton University Press, Princeton, N.J., 1955, p. 155.
3. Luke, Y.L. *The Special Functions and Their Approximations Vol II*. Academic Press, New York and London, 1969, p. 196.
4. Shanks, D. Non-linear transformations of divergent and slowly convergent sequences. *J. Math. Phys.* **34** (1955), 1.
5. Takenaga, R. On the evaluation of the incomplete gamma function. *Math. Computation* **20** (Oct. 1966), 606.

Algorithm

```

FUNCTION GAMINC (A,X1,X2,GAM)
C
C COMPUTE THE DIFFERENCE BETWEEN TWO MODIFIED INCOMPLETE
C GAMMA FUNCTIONS FOR (A,X1) AND (A,X2) THEN MULTIPLY BY
C EXP(X1). THAT IS, COMPUTE THE INTEGRAL OF ABS(X)**(A-1.)
C *EXP(X1-X) FROM X1 TO X2. IF X1 .GT. X2, THEN X1-X2 MUST
C BE .LE. EXPLIM.
C EXPLIM CAN BE A MACHINE DEPENDENT CONSTANT WHICH PREVENTS
C EXPONENTIATION OVER- AND UNDERFLOWS. IT IS USED HERE TO
C SUPPRESS THE CALCULATION OF MIGAM(A,X2) WHEN THE VALUE OF
C MIGAM(A,X2) IS INSIGNIFICANT. THIS USAGE REQUIRES X2 +
C EXPLIM .GE. X1. (MIGAM IS AN ABBREVIATION FOR MODIFIED IN-
C COMPLETE GAMMA FUNCTION.)
C GAM IS THE COMPLETE GAMMA FUNCTION OF A SUPPLIED BY THE
C CALLING PROGRAM.
C
C FOR X .GT. 5., GAM-MIGAM(A,X) IS COMPUTED WITH A CONTINUED
C FRACTION APPROXIMATION. FOR ABS(X) .LE. 1.0, THE INTEGRAL
C IS TRANSFORMED AND EXP(Q) IS APPROXIMATED WITH A CHEBYSHEV
C SERIES SO THAT THE NEW INTEGRAL MAY BE DONE ANALYTICALLY.
C FOR X .GT. -12. AND X .LT. 5. (ABS(X) .GT. 1.0), A CONTIN-
C UED FRACTION APPROXIMATION IS USED. FINALLY FOR X .LE.
C -12., THE ASYMPTOTIC EXPANSION IS USED.
C
C SGN IS A SWITCH WHICH, IF NONZERO, INDICATES WHETHER GAM
C SHOULD BE ADDED OR SUBTRACTED FROM AN INTERMEDIATE RESULT.
C
      DATA EXPLIM/20./
      Z=X1
      SGN=0.
      TIM=-1.
      EXPDIF=1.0
5     IF (Z .NE. 0.) GO TO 10
      GAM1=0.
      SGN=SGN+TIM
      GO TO 40
10    IF (Z .LE. 5.) GO TO 20
C USE EQUATION 10.
      GAM1=-EXPDIF*Z**A/(Z+(1.-A)/(1.+1./((Z+(2.-A)/(1.+2.
1     /((Z+(3.-A)/(1.+3./((Z+1.7)))))))
      GO TO 40
20    AZ=ABS(Z)
      IF (Z .LE. -12.) GO TO 30
      SGN=SGN+TIM
C USE EQUATION 17.
      IF (AZ .LE. 1.) GAM1=EXPDIF*Z/A *AZ**A*(A-1.)
1     *(1. +Z/(A+1.) *(.99999999+Z/(A+2.)
2     *(.99999999 +Z/(A+3.) *(1.000008+Z/(A+4.)
3     *(1.000005 +Z/(A+5.) *(.9994316+Z/(A+6.)
4     *(.9995587 +Z/(A+7.) *(1.031684+Z/(A+8.)
5     *(1.028125))))))
C USE EQUATIONS 11 AND 12. EVALUATION MUST BE DONE
C IN DOUBLE PRECISION IF COMPUTER HAS 32 OR FEWER BITS

```

```

C PER WORD. ON AN IBM 360, D. P. EVALUATION IS FORCED
C BY THE D. P. CONSTANTS IN CONTINUATION CARD 9.
      IF (AZ .GT. 1.) GAM1=EXPDIF*Z/A *AZ**A*(A-1.)
1     /(1.- A' *Z/(A *(A+ 1.+ Z/(A+ 2.)
2     *(1.- (A+1.) *Z/(A+ 2.) *(A+ 3.+2. *Z/(A+ 4.)
3     *(1.- (A+2.) *Z/(A+ 4.) *(A+ 5.+3. *Z/(A+ 6.)
4     *(1.- (A+3.) *Z/(A+ 6.) *(A+ 7.+4. *Z/(A+ 8.)
5     *(1.- (A+4.) *Z/(A+ 8.) *(A+ 9.+5. *Z/(A+10.)
6     *(1.- (A+5.) *Z/(A+10.) *(A+11.+6. *Z/(A+12.)
7     *(1.- (A+6.) *Z/(A+12.) *(A+13.+7. *Z/(A+14.)
8     *(1.00150-A*8.95E-5 +Z*(-.0337062+A*.0004182
9     +Z*(.000999294-A*.000104103))))))
      A )))) )))) )))) ))))
      GO TO 40
C USE EQUATION 18 AND SHANK-S E1 PROCESS ONCE.
30    GAM1=-EXPDIF*AZ**A*(A-1.)*(1.- (A-1.)*(1.- (A-2.)*
1     (1.- (A-3.)*(1.- (A-4.)*(1.- (A-5.)/(Z-A+6.))
2     /Z)/Z)/Z)/Z)
40    IF (TIM .GT. 0.) GO TO 55
      GAMINC=GAM1
      IF (ABS(X1-X2) .GT. EXPLIM) GO TO 50
C IF TRUE, CONTRIBUTION AT X2 IS .LT. 1.E-7 *(CONTR AT X1),
C PROVIDED X2 .GT. X1.
      Z=X2
      EXPDIF=EXP(X1-X2)
      TIM=1.
      GO TO 5
50    GAM1=0.
55    GAMINC=GAM1-GAMINC
      IF (SGN .NE. 0.) GAMINC=GAMINC-SIGN(GAM*EXP(X1),SGN)
      RETURN
      END

```